Media Player Visualizations


by Chris Bell


Windows Media Player 7 added support for visualizations, add-ins that allow developers to produce graphical effects that change their behaviour in time with the music. Media Player 9 extended their power further. This article provides a brief introduction to the world of visualizations and shows how to implement them in Delphi.


I’ve always liked the visual imagery provided by Media Player’s visualizations. When I realised that is was possible to create my own plug-ins like this I got quite excited, when I read the MSDN documentation telling me that these could only be written in Visual C++ I got somewhat annoyed. Since they appeared to be just COM DLLs, why couldn’t I use Delphi? The naughty trick MS has played here is not to provide a type library for the required interface. 


So how to go about developing a Delphi visualization. The first thing to do was download the Media Player SDK, which provides a wizard for Visual C++. After downloading this and installing the wizard into Visual C++, I was ready to go. I created a new visualization and had a look around. It turns out there are two interfaces that are of interest, IWMPEffects and IWMPEffects2. IWMPEffects is required for visualizations in Media Player 7 and after, IWMPEffects2 inherits from this and can be used with visualizations in Media Player 9. You must implement the first interface but not necessarily the second one. I’ll concentrate on the first interface for the moment, since it’s easier to deal with. Let’s first see what it looks like in C++


MIDL_INTERFACE("D3984C13-C3CB-48e2-8BE5-5168340B4F35")


    IWMPEffects : public IUnknown


    {


    public:


        virtual /* [helpstring][local] */ HRESULT STDMETHODCALLTYPE Render( 


            /* [in] */ TimedLevel *pLevels,


            /* [in] */ HDC hdc,


            /* [in] */ RECT *prc) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE MediaInfo( 


            /* [in] */ LONG lChannelCount,


            /* [in] */ LONG lSampleRate,


            /* [in] */ BSTR bstrTitle) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE GetCapabilities( 


            /* [out] */ DWORD *pdwCapabilities) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE GetTitle( 


            /* [out] */ BSTR *bstrTitle) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE GetPresetTitle( 


            /* [in] */ LONG nPreset,


            /* [out] */ BSTR *bstrPresetTitle) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE GetPresetCount( 


            /* [out] */ LONG *pnPresetCount) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE SetCurrentPreset( 


            /* [in] */ LONG nPreset) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE GetCurrentPreset( 


            /* [out] */ LONG *pnPreset) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE DisplayPropertyPage( 


            /* [in] */ HWND hwndOwner) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE GoFullscreen( 


            /* [in] */ BOOL fFullScreen) = 0;


        virtual /* [helpstring] */ HRESULT STDMETHODCALLTYPE RenderFullScreen( 


            /* [in] */ TimedLevel *pLevels) = 0;


        


    };





Without wanting to start a religious war, it has to be said this is damn ugly. C++ has no idea about interfaces, so this is a class with lots of virtual abstract functions. Delphi also hides those nasty HRESULTs with the handy safecall calling convention. 


So the first thing to do is to convert this into Delphi code. When I initially did this conversion I did it all by hand, but I thought for the sake of this article I’d give Darth a go (http://delphi-jedi.org/team_darth_home). This is a C to Pascal header conversion tool originally developed by Dr Bob, now taken over by Alan Moore. Unfortunately it didn’t really make any kind of successful attempt at converting this code, but I guess that’s because it was only ever meant to deal with simple C-style functions and not bizarre C++ interfaces.


So after some fiddling around, I came up with this


IWMPEffects = interface(IUnknown)


  	['{D3984C13-C3CB-48e2-8BE5-5168340B4F35}']


    procedure Render(var pLevels : TimedLevel; hdc : HDC; var prc : TRect); safecall;


    procedure MediaInfo(lChannelCount : longint; lSampleRate : longint; bstrTitle : WideString); safecall;


    procedure GetCapabilities(var pdwCapabilities : DWORD); safecall;


    procedure GetTitle(var bstrTitle : WideString); safecall;


    procedure GetPresetTitle(nPreset : LongInt; var bstrPresetTitle : WideString); safecall;


    procedure GetPresetCount(var pnPresetCount : LongInt); safecall;


    procedure SetCurrentPreset(nPreset : LongInt); safecall;


    procedure GetCurrentPreset(var pnPreset : LongInt); safecall;


    procedure DisplayPropertyPage(hwndOwner : HWND); safecall;


    procedure GoFullscreen(fFullScreen : BOOL); safecall;


    procedure RenderFullScreen(var pLevels : TimedLevel); safecall;


  end; 





So what does this interface do? The COM DLL that will implement your visualization can actually implement several visualizations. GetTitle should return the name of your collection of visualizations. GetPresetCount should return the number of visualizations in the DLL, whilst GetPresetTitle should return the name of each visualization. SetCurrentPreset tells you which of your visualizations to display. Finally, the most important function to implement is the Render function. This allows you to get artistic and draw your visualization on screen based on the sound data currently being played. The TimedLevel record (See below) provides the information about the sound data. This includes stereo frequency information and stereo waveform information.


TimedLevel = record


    frequency : array [0..1, 0..SA_BUFFER_SIZE-1] of byte;


    waveform  : array [0..1, 0..SA_BUFFER_SIZE-1] of byte;


    state : integer;


    timeStamp : int64;


  end;





So, after creating a COM DLL and implementing the required interface, I was upset to not see it appear in Media Player. Unfortunately there’s a little more work required before Media Player will take notice of your visualization. To be fair Media Player has no way of knowing the DLL is for its use. Looking through the code generated by the Visual C++ wizard gave no obvious indication of what else needed to be added to the registry. This is where the exceedingly useful RegMon tool came in handy. This, along with various other tools, can be found at the SysInternals website (http://www.sysinternals.com/). Using this whilst registering the demo DLL showed those pesky registry entries we need. As an aside, it also shows that Media Player 9 looks at the HKEY_CURRENT_USER\Software\Microsoft\MediaPlayer\Preferences\UsageTracking registry key about 100 times a second… 


Our new DllRegisterServer thus looked like


function DllRegisterServer: HResult;


var


  LReg : TRegistry;


begin


  Result := S_OK;


  try





    // standard registration


    ComServer.UpdateRegistry(True);





    // register visualization stuff


    LReg := TRegistry.Create;


    try


      LReg.RootKey := HKEY_LOCAL_MACHINE;


      if LReg.OpenKey('Software\Microsoft\MediaPlayer\Objects\Effects', True) then


      begin


        if LReg.OpenKey('Doogal', True) then


        begin


          if LReg.OpenKey('Properties', True) then


          begin


            LReg.WriteString('classid', GUIDToString(Class_Doogal));


            LReg.WriteString('description', 'my description');


            LReg.WriteString('name', 'doogal');


          end;


        end;


      end;


    finally


      LReg.Free;


    end;





    PostMessage(HWND_BROADCAST, RegisterWindowMessage('WMPlayer_PluginAddRemove'), 0, 0);


  except


    Result := E_FAIL;


  end;


end; 





Note also the final PostMessage that tells Media Player, if it’s currently running, that a new visualization has been registered


Conclusion


This has been a brief introduction to Media Player Visualizations. There are lots of things I haven’t covered, such as the IWMPEffects2 interface available in Media Player 9 that provides full access to the Media Player object model, therefore enabling much more powerful development opportunities. It also crossed my mind that it would be useful to create a Delphi wizard for creating these visualizations or a class wrapper round some of the nasty WinAPI stuff. I’ve not got round to doing wither of these things, but if anybody takes up the challenge, I’d love to hear from them


I hope this has been of interest to people. You can find a demo visualization on my website at http://www.doogal.co.uk/mplayer.php


